IEEE Std. 1394-1995 Electrical Issues

Discussion Topics

- Spice modeling 1394 connectors and cables
- Data+Strobe signaling problems

Spice Modeling Complex Transmission Lines



Nonlinear sources permit modeling arbitrary combinations of excitation currents into the transmission line model

Spice Modeling Differential Pair Transmission Lines



Differential Mode Connector Model

Differential Mode Connector + Transmission Line Model



Common Mode Connector Model

Common Mode Connector + Transmission Line Model



Common Mode Results



Experimental TDR Data



Spice Simulated TDR Data

Differential Mode Results



Experimental TDR Reflectivity



Spice TDR Reflectivity Simulation

S400 K28.5 over 4.5 meters + 2 connectors



Tr/f = 100 pS

Experimental Data



Tr/f = 1.2 nS

Spice Simulation

S400 K28.5 over 20 inches + 2 connectors



Spice Simulation

Tr/f = 1.2 nS



Spice Simulation

Tr/f = 100 pS

Copper Interconnect Subcircuit Spice Deck

```
.SUBCKT COPPER_CNT 1 2 3 4
* MODEL OF TWO 1394-1995 CONNECTORS + 4.5 METERS OF CABLE
* PINS 1 AND 2 ARE ONE SIDE, PINS 3 AND 4 ARE THE OTHER SIDE
* PINS 1 AND 3 ARE + DIFF VOLTAGE, PINS 2 AND 4 ARE - DIFF VOLTAGE

*                       DEFINE AMMETERS:
VX1 1 5 DC 0VOLTS
VX2 2 6 DC 0VOLTS
VX3 3 7 DC 0VOLTS
VX4 4 8 DC 0VOLTS

*                       DEFINE VOLTAGE GENERATORS FOR PINS:
BVPIN1 5 0 V=V(11)+V(9)/2
BVPIN2 6 0 V=V(11)-V(9)/2
BVPIN3 7 0 V=V(16)+V(10)/2
BVPIN4 8 0 V=V(16)-V(10)/2

*                       DEFINE CURRENT SOURCES FOR THE TRANSMISSION LINES:
BIDIFF12 0 9 I=I(VX1)-I(VX2)
BIDIFF34 0 10 I=I(VX3)-I(VX4)
BICOM12 0 11 I=I(VX1)+I(VX2)
BICOM34 0 16 I=I(VX3)+I(VX4)

*                       DEFINE DIFFERENTIAL MODE CONNECTORS+CABLE
*                                               CONNECTOR: 1 INCH TRANS LINE + CAPACITOR:
.MODEL LOSSYDIFFSHORT LTRA R=0.99 G=0 L=13.56E-9 C=1.323E-12 LEN=1
ODIFF12 9 0 109 0 LOSSYDIFFSHORT
CCONN1 109 0 0.65PF

ODIFF34 10 0 110 0 LOSSYDIFFSHORT
CCONN2 110 0 0.65PF

*DIFF TRANSMISSION LINE
*                                               MODEL FOR 4.5 METER DIFFERENTIAL MODE TRANSMISSION LINE:
*                                               MODEL HAND TUNED AGAINST 5.8 DB LOSS AT 400 MHZ, BUT FIGURED FOR 200 MHZ
.MODEL LOSSYDIFF LTRA R=0.99 G=0 L=13.56E-9 C=1.323E-12 LEN=177
ODIFF 109 0 110 0 LOSSYDIFF

*                       DEFINE COMMON MODE CONNECTORS+CABLE
*                                               CONNECTOR: 1 INCH TRANS LINE + INDUCTOR-CAPACITOR-INDUCTOR:
.MODEL LOSSYCOMMONSHORT LTRA R=.30 G=0 L=4E-9 C=4.5E-12 LEN=1
OCOM12 11 0 111 0 LOSSYCOMMONSHORT
LCONN1 111 12 3.2NH
CCONN3 12 0 3PF
LCONN2 12 13 3NH

OCOM34 16 0 116 0 LOSSYCOMMONSHORT
LCONN3 15 116 3.2NH
CCONN4 15 0 3PF
LCONN4 14 15 3NH

*COMMON MODE TRANSMISSION LINE
*                                               MODEL FOR 4.5 METER COMMON MODE TRANSMISSION LINE:
*                                               MODEL HAND TUNED AGAINST 5.8 DB LOSS AT 400 MHZ, BUT FIGURED FOR 200 MHZ
.MODEL LOSSYCOMMON LTRA R=.30 G=0 L=4E-9 C=4.5E-12 LEN=177
OCOM 13 0 14 0 LOSSYCOMMON

.ENDS
```

Spice Model Caveats

- Lossy cable model is frequency independent, real cables are frequency dependent (e.g., skin depth effects)
- Common mode losses tuned for 200 MHz, skew creates common mode signals in the GHz range
- Connector parasitics may not correlate with the connector physics, values were tuned against the cable impedances
- Cables and connectors have significant differential and common mode cross-talk -- this model does not handle cross-talk

Data/Strobe Signaling Analysis







Worst case cable attenuation



Worst case short cable received signal: S400, 200ps Tr/f, 18” long

Minimum Rise Time Study



Worst case short cable received signal: S400, 200ps Tr/f, 18” long



Acceptable case short cable received signal: S400, 500ps Tr/f, 18” long

Receiver Sensitivity



4.5 meters, 500 ps rise/fall, worst case launch voltage (172 mV) and cable losses

P1394a Recommendations

- Add minimum rise/fall time (10-90%) requirement of 500 ps for S400
- Change receiver sensitivity for S400 to 60 mV differential (120 mV differential peak to peak)